


N 73:27579

**NASA TECHNICAL
MEMORANDUM**



NASA TM X-2836

NASA TM X-2836

**CASE FILE
COPY**

**DOPEX-1D2C - A ONE-DIMENSIONAL,
TWO-CONSTRAINT RADIATION
SHIELD OPTIMIZATION CODE**

by Gerald P. Labti

Lewis Research Center

Cleveland, Ohio 44135

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • WASHINGTON, D. C. • JULY 1973

1. Report No. NASA TM X-2836		2. Government Accession No.		3. Recipient's Catalog No.	
4. Title and Subtitle DOPEX-1D2C - A ONE-DIMENSIONAL, TWO-CONSTRAINT RADIATION SHIELD OPTIMIZATION CODE				5. Report Date July 1973	
				6. Performing Organization Code	
7. Author(s) Gerald P. Lahti				8. Performing Organization Report No. E-7421	
				10. Work Unit No. 503-05	
9. Performing Organization Name and Address Lewis Research Center National Aeronautics and Space Administration Cleveland, Ohio 44135				11. Contract or Grant No.	
				13. Type of Report and Period Covered Technical Memorandum	
12. Sponsoring Agency Name and Address National Aeronautics and Space Administration Washington, D.C. 20546				14. Sponsoring Agency Code	
15. Supplementary Notes					
16. Abstract A one-dimensional, two-constraint radiation shield weight optimization procedure and a computer program, DOPEX-1D2C, is described. DOPEX-1D2C uses the steepest-descent method to alter a set of initial (input) thicknesses of a spherical shield configuration to achieve a minimum weight while simultaneously satisfying two dose-rate constraints. The code assumes an exponential dose-shield thickness relation with parameters specified by the user. Code input instruction, a FORTRAN-IV listing, and a sample problem are given. Typical computer time required to optimize a seven-layer shield is less than 1/2 minute on an IBM 7094.					
17. Key Words (Suggested by Author(s)) Radiation shield Weight optimization				18. Distribution Statement Unclassified - unlimited	
19. Security Classif. (of this report) Unclassified		20. Security Classif. (of this page) Unclassified		21. No. of Pages 31	
				22. Price* \$3.00	

* For sale by the National Technical Information Service, Springfield, Virginia 22151

DOPEX-1D2C - A ONE-DIMENSIONAL, TWO-CONSTRAINT RADIATION SHIELD OPTIMIZATION CODE

by Gerald P. Lahti

Lewis Research Center

SUMMARY

A one-dimensional, two-constraint radiation shield weight optimization procedure and a computer program, DOPEX-1D2C, is described. DOPEX-1D2C uses the steepest-descent method to alter a set of initial (input) thicknesses of a spherical shield configuration to achieve a minimum weight while simultaneously satisfying two dose-rate constraints. The code assumes an exponential dose-shield thickness relation with parameters specified by the user. Code input instruction, a FORTRAN-IV listing, and a sample problem are given. Typical computer time required to optimize a seven-layer shield is less than 1/2 minute on an IBM 7094.

INTRODUCTION

One-dimensional, single-constraint shield weight optimization procedures have been established for radiation shield design (refs. 1 and 2). Optimization procedures such as these take an initial configuration specified by the user and alter the thickness of each layer to achieve a minimum weight while simultaneously satisfying a single dose constraint.

One method used to seek the minimum-weight configuration from a given starting configuration is that of steepest descent. Bernick (ref. 3) developed an algorithm to optimize one-dimensional shields with one constraint by utilizing the steepest-descent method and programmed it as the OPEX code (ref. 4). OPEX was later reprogrammed for spherical geometry and reinterpreted as the OPEX-II code by Lahti (ref. 5). The steepest-descent strategy was extended in theory by Lahti (ref. 6) to the multidimensional, multiconstraint situation. A special case of this multidimensional optimization was programmed as the DOPEX code (ref. 7). DOPEX considered the special case where doses in each principal direction (i. e., R-axis and plus and minus Z-axis of a

right-circular cylinder) are dependent only on shield thicknesses in that direction. DOPEX might be considered to be a one-dimensional, one-constraint procedure coupled through the weight equation to operate in a two-dimensional, two- or three-constraint situation.

The present report describes the next step in the evolution of multiconstraint optimization strategy, the DOPEX-1D2C code. DOPEX-1D2C is a one-dimensional, two-constraint optimization code and is presently coded for spherical geometry. The optimization algorithm and equations are those derived in reference 6, but because of an instability, a new algorithm for obtaining an initial configuration that would meet the input dose constraint had to be derived. Included in this report are derivations, data input instructions, a FORTRAN-IV code listing, and a sample problem for DOPEX-1D2C.

DOSE-THICKNESS RELATION

The DOPEX-1D2C code considers two dose rates $D_1(\bar{t})$ and $D_2(\bar{t})$ at some reference point in space. These dose rates are functions of layer thicknesses \bar{t} . Each $D_m(\bar{t})$, $m = 1$ or 2 , is in turn defined as

$$D_m(\bar{t}) = \sum_{i=1}^{IMAX_m} D_{im}(\bar{t})$$

where

$D_{im}(\bar{t})$ i^{th} component of m^{th} total dose rate (e.g., dose due to capture gammas from the first shield layer or dose due to inelastic gammas from the last shield layer, etc.)

$D_m(\bar{t})$ total dose rate, $m = 1$ or 2 (e.g., D_1 might be total neutron dose rate and D_2 might be total gamma dose rate)

$IMAX_m$ number of dose-rate components in D_m

\bar{t} set of t_j , the thickness of the j^{th} region

Each dose-rate component $D_{im}(\bar{t})$ is further assumed to be of the form

$$D_{im}(\bar{t}) = C_{im} \exp \left(- \sum_{j=1}^N \mu_{ijm} t_j \right)$$

where

C_{im} fitted parameter

N number of regions (thicknesses) in the configuration

t_j thickness of the j^{th} region, cm

μ_{ijm} "attenuation coefficient," cm^{-1}

This formulation is similar to that used in the OPEX-II and DOPEX codes and will not be detailed further in this report. For additional information on how to obtain these coefficients, the reader is referred to references 5 and 7.

WEIGHT-MINIMIZATION PROCEDURE

The procedure for obtaining the minimum-weight configuration by the method of steepest descent is presented in this section. Details of derivations can be found in reference 6.

The problem to be solved is that of minimizing the system weight w (a function of thickness t_j) while simultaneously satisfying two dose-rate constraints D_1 and D_2 (also a function of t_j). Additional constraints to be imposed are that the thickness t_j must be nonnegative and that certain thicknesses may be constrained to constant thickness. The problem, then, is to minimize $w(t_1, t_2, \dots, t_n)$ with the following constraints:

$$(1) D_1(\bar{t}) = D_1^0$$

$$(2) D_2(\bar{t}) = D_2^0$$

$$(3) t_j \geq 0$$

$$(4) t_l \text{ is constant for any desired values of } l.$$

Constraint 3 ensures a physically meaningful solution. Constraint 4, fixed thickness, is useful if it is desired that some thicknesses be kept from changing during the course of the calculation (e.g., the reactor core size and the reflector thickness).

An n -dimensional Euclidean vector space with Cartesian coordinates, t_1, t_2, \dots, t_N is defined. The following vectors are defined on this space:

$$\bar{\mathbf{t}} = \begin{pmatrix} t_1 \\ t_2 \\ \vdots \\ t_N \end{pmatrix} \quad \bar{\mathbf{g}}(\bar{\mathbf{t}}) = \begin{pmatrix} \frac{\partial w(\bar{\mathbf{t}})}{\partial t_1} \\ \frac{\partial w(\bar{\mathbf{t}})}{\partial t_2} \\ \vdots \\ \frac{\partial w(\bar{\mathbf{t}})}{\partial t_N} \end{pmatrix} \quad \bar{\mathbf{a}}_1(\bar{\mathbf{t}}) = \begin{pmatrix} \frac{\partial D_1(\bar{\mathbf{t}})}{\partial t_1} \\ \frac{\partial D_1(\bar{\mathbf{t}})}{\partial t_2} \\ \vdots \\ \frac{\partial D_1(\bar{\mathbf{t}})}{\partial t_N} \end{pmatrix} \quad \bar{\mathbf{a}}_2(\bar{\mathbf{t}}) = \begin{pmatrix} \frac{\partial D_2(\bar{\mathbf{t}})}{\partial t_1} \\ \frac{\partial D_2(\bar{\mathbf{t}})}{\partial t_2} \\ \vdots \\ \frac{\partial D_2(\bar{\mathbf{t}})}{\partial t_N} \end{pmatrix}$$

The unit vector $\hat{\mathbf{u}}(\bar{\mathbf{t}})$ (see ref. 6 for derivation)

$$\hat{\mathbf{u}}(\bar{\mathbf{t}}) = \frac{-\bar{\mathbf{g}} + \frac{\bar{\mathbf{a}}_1 \cdot \bar{\mathbf{g}}}{\bar{\mathbf{a}}_1 \cdot \bar{\mathbf{a}}_1} \bar{\mathbf{a}}_1 + \frac{\bar{\mathbf{a}}_2 \cdot \bar{\mathbf{g}}}{\bar{\mathbf{a}}_2 \cdot \bar{\mathbf{a}}_2} \bar{\mathbf{a}}_2}{\left[\bar{\mathbf{g}} \cdot \bar{\mathbf{g}} - \frac{(\bar{\mathbf{a}}_1 \cdot \bar{\mathbf{g}})^2}{\bar{\mathbf{a}}_1 \cdot \bar{\mathbf{a}}_1} - \frac{(\bar{\mathbf{a}}_2 \cdot \bar{\mathbf{g}})^2}{\bar{\mathbf{a}}_2 \cdot \bar{\mathbf{a}}_2} \right]^{1/2}}$$

points in the direction of greatest weight decrease (steepest descent) along a hyperplane tangent to the hypersurface described by the equations

$$D_1(\bar{\mathbf{t}}) = D_1^0 \quad \text{the first dose constraint}$$

and

$$D_2(\bar{\mathbf{t}}) = D_2^0 \quad \text{the second dose constraint}$$

Components of $\hat{\mathbf{u}}(\bar{\mathbf{t}})$ represent increments of thickness to be added to $\bar{\mathbf{t}}$ in order to approach a minimum weight.

The optimization algorithm, which is similar to that used in OPEX, OPEX II, and DOPEX, is as follows:

- (1) The unit vector $\hat{\mathbf{u}}(\bar{\mathbf{t}})$ is calculated for the present value of $\bar{\mathbf{t}}$.
- (2) A fraction, f , of $\hat{\mathbf{u}}$ is added to $\bar{\mathbf{t}}$. The fraction f is an input parameter. (A value of $f = 1.0$ has given satisfactory results.) That is,

$$\bar{t} = \bar{t} + \hat{f}u(\bar{t})$$

(3) The new set of thicknesses \bar{t} generally does not return the correct dose constraints because of the nonlinear dose-thickness relation. A first-order correction \overline{dt} is applied to \bar{t} to restore the dose constraints. This correction (derived in appendix A as eq. (A12)) is

$$\overline{dt} = - \frac{[D_2^*(\nabla D_1 \cdot \nabla D_2) - D_1^* \nabla^2 D_2] \nabla D_1 + [D_1^*(\nabla D_1 \cdot \nabla D_2) - D_2^* \nabla^2 D_1] \nabla D_2}{(\nabla D_1 \cdot \nabla D_2)^2 - \nabla^2 D_1 \nabla^2 D_2}$$

where D_m^* is the difference, $D_m(\bar{t}) - D_m^0$. Finally, then, in this step, \bar{t} is incremented by \overline{dt} to obtain a new \bar{t} which satisfies dose constraints.

(4) The system weight is calculated for this new \bar{t} . If the change in weight (between that calculated in the present iteration and that in the past iteration) is less than some prescribed ϵ , then the algorithm stops. Otherwise, steps 1 to 4 are repeated.

INITIALIZATION PROCEDURE

In general, an initial (or estimated) configuration will not satisfy dose constraints. Therefore, provision must be made to obtain some initial feasible solution before the optimization can proceed. The method used in OPEX-II and DOPEX turned out to be unstable for the present two-constraint case. Therefore, a new initialization algorithm was developed. Details of this derivation appear in Appendix A.

The initialization algorithm proceeds as follows:

(1) Calculate dose rates $D_1(\bar{t})$ and $D_2(\bar{t})$ and compare with the required dose constraints D_1^0 and D_2^0 .

(2) If $D_m(\bar{t}) > 2D_m^0$ (where $m = 1, 2$), set $D_m^* = 0.5 D_m(\bar{t})$.

If $D_m(\bar{t}) < 0.5 D_m^0$ (where $m = 1, 2$), set $D_m^* = 2D_m(\bar{t})$.

If $0.5 < [D_m(\bar{t})/D_m^0] < 2.0$ (where $m = 1, 2$), set $D_m^* = D_m(\bar{t}) - D_m^0$.

(3) Using the differences D_1^* and D_2^* calculated in step 2, use equation (A12) to calculate \overline{dt} and a new $\bar{t} = \bar{t} + \overline{dt}$ which will result in dose rates of approximately D_1^* and D_2^* .

(4) Calculate $D_m(\bar{t})$ and

$$\left| \frac{D_m(\bar{t}) - D_m^0}{D_m^0} \right|$$

and compare to some convergence parameter ϵ . If

$$\left| \frac{D_m(\bar{t}) - D_m^0}{D_m^0} \right| < \epsilon$$

for $m = 1$ and 2 , the algorithm is complete. If

$$\left| \frac{D_m(\bar{t}) - D_m^0}{D_m^0} \right| > \epsilon$$

for $m = 1$ or 2 , go to step 2.

The present procedure has been uniformly convergent on all problems tested to date. The arbitrary range (factor of 0.5 to 2.0), assumed for the first-order difference equation to be valid, has so far been demonstrated to be adequate. The user may, of course, change this parameter in subroutine INIT if desired.

DOPEX-1D2C CODE

In this section, the details of the DOPEX-1D2C code are presented. Included are (1) an overview of the code, (2) a flow chart for data input, and (3) a sample problem and sample problem output. A complete FORTRAN IV listing appears in appendix B.

Overview of DOPEX-1D2C Code

Objective. - DOPEX-1D2C is a radiation shield optimization code which obtains a minimum-weight configuration for a layered, spherical shell arrangement while simultaneously satisfying two dose-rate constraints. This code alters layered configurations. It will not add shield layers, but it may delete layers.

Method. - The method of steepest descent is used in an algorithm designed to alter an input configuration while simultaneously satisfying dose-rate constraints.

Required input data. - The user must provide an initial geometric configuration (layer thicknesses and densities of materials) and all data to be used in an empirical equation which relates changes in dose rates with changes in thickness.

Computer and language used. - DOPEX-1D2C is written in ANSI-standard FORTRAN-IV and has been compiled and run on IBM 7094 and IBM 360/67 computers.

Restrictions in problem size. - DOPEX-1D2C is presently set up for a maximum of 25 regions and each of the two dose constraints is limited to 25 components. These limits could be increased as DOPEX-1D2C presently requires a small fraction of the 32K IBM 7094 available core.


Typical machine time. - A typical DOPEX-1D2C run requires less than 1/2 minute on an IBM 7094.


Precautions. - The final configuration predicted by DOPEX-1D2C is only as good as the empirical dose-thickness equation used in the calculation. The user is advised to check the predicted dose rate for the final configuration by some more exact method.

Other geometries. - Other geometries can be used by rewriting subroutine WEIGHT to suit.


More constraints. - DOPEX-1D2C is presently written to handle only two constraints. However, equations for more constraints have been derived and could be incorporated if deemed necessary.

Flow Chart for Data Input

The following is a flow chart for data input to DOPEX-1D2C. The symbol  means read a card of set of cards as described.



	Card column	Variable	Description	FORMAT
	21-30	EPSD	Convergence criterion for initial dose (typically, 0.001)	
	31-40	CON	Fractional step size on \hat{u} (typically, about 1.0)	
3	-----	T(J)	Thicknesses of each region; NREG values required	7E 10.4
4	-----	RHO(J)	Densities of each region; NREG values required	7E 10.4
5	-----	NB(J)	Constraint flags; NREG values required. NB(J) = 0 constrains j^{th} region to constant thickness NB(J) = 1 allows j^{th} region to change	25I1
6	1-5	IMAX(1)	Dose rate control card 1 Number of dose-rate components in first dose-rate constraint	I5, E 10.4
	6-15	DSTAR(1)	Value of first dose-rate constraint	
7	-----	EMU(I, J)	Attenuation coefficients, μ_{ij} . For the first constraint, read a new card (or set of cards) with IMAX(1) values of μ_{ij} , NREG cards or sets of cards are required.	7E 10.4
8		D(I, 1)	Dose-rate components D_i for first constraint; IMAX(1) values required	7E 10.4



Sample Problem

A sample problem concerning a reactor, a molybdenum reflector, and a shield consisting of seven alternating layers of lithium hydride and tungsten is illustrated in figure 1. This particular arrangement is the same as that used in the OPEX-II code and was optimized in that case for a single dose constraint of 2 mrem/hour at a distance of 20 meters.

As a test of the DOPEX-1D2C code, this same geometric layout is optimized for constraints of 0.2527-mrem/hour neutron dose rate and 1.7473-mrem/hour total gamma dose rate. These values were predicted by the single-constraint optimization calculation (OPEX-II). The present case, with two constraints, should result in a layout and weight similar to that predicted by the one-constraint optimization. The present case will be initiated at the same configuration as was the OPEX-II calculation.

The results of this sample problem are given in tables I to IV. Table I lists region descriptions, densities, initial thicknesses, and the thicknesses predicted by OPEX-II and by DOPEX-1D2C. Table II lists dose rates by component. Within the DOPEX-1D2C calculation, the neutron dose (item 1 in table II) is considered to be the first constraint and the total gamma dose (sum of items 2 to 12 in table II) is considered to be the second constraint. The configurations calculated by OPEX-II and DOPEX-1D2C are similar. Although the thicknesses are not identical, the system weight, 3×10^7 grams, is. It has been observed that minimum-weight configurations for thick shields, as considered herein, can exhibit considerable local variations. This simply demonstrates that the solution given by OPEX-II or DOPEX-1D2C is not a unique one. Table III lists the complete DOPEX-1D2C output for this case. Table IV gives the coefficients used in this run.

Lewis Research Center,
National Aeronautics and Space Administration,
Cleveland, Ohio, May 3, 1973,
503-05.

APPENDIX A

RECALCULATION OF \bar{t} TO MEET CONSTRAINTS

The problem here is the following: given a vector \bar{t} calculate

$$D_1^*(\bar{t}) = D_1(\bar{t}) - D_1^0$$

$$D_2^*(\bar{t}) = D_2(\bar{t}) - D_2^0$$

where $D^*(\bar{t})$ is the difference between the dose rate as calculated using \bar{t} (i.e., $D(\bar{t})$) and the required dose constraint D^0 . It is required that $D^*(\bar{t}) = 0$ to satisfy dose-rate constraints. To obtain a feasible solution for \bar{t} (i.e., one where $D_1^* = 0$ and $D_2^* = 0$), assume the first-order correction is valid. That is,

$$D_1^*(\bar{t} + d\bar{t}) = D_1^*(\bar{t}) + \nabla D_1^*(\bar{t}) \cdot d\bar{t}$$

and

$$D_2^*(\bar{t} + d\bar{t}) = D_2^*(\bar{t}) + \nabla D_2^*(\bar{t}) \cdot d\bar{t}$$

For the present case we require $d\bar{t}$ for which the residuals

$$D_1^*(\bar{t} + d\bar{t}) = 0$$

and

$$D_2^*(\bar{t} + d\bar{t}) = 0$$

Also, to minimize perturbations on the system, we also will require that $|d\bar{t}|$ be a minimum.

The problem is now restated as

$$\text{Minimize } d\bar{t} \cdot d\bar{t} \tag{A1}$$

subject to

$$D_1^*(\bar{t}) + \nabla D_1^*(\bar{t}) \cdot d\bar{t} = 0 \tag{A2}$$

and

$$D_2^*(\bar{t}) + \nabla D_2^*(\bar{t}) \cdot \bar{dt} = 0 \quad (A3)$$

To solve this, form the Lagrangian \mathcal{L} by taking

$$\mathcal{L} = \bar{dt} \cdot \bar{dt} + \gamma_1 [D_1^*(\bar{t}) + \nabla D_1^*(\bar{t}) \cdot \bar{dt}] + \gamma_2 [D_2^*(\bar{t}) + \nabla D_2^*(\bar{t}) \cdot \bar{dt}]$$

where γ_1 and γ_2 are multipliers to be determined. Equivalently,

$$\mathcal{L} = \bar{dt} \cdot \bar{dt} + \gamma_1 [D_1^*(\bar{t}) + \nabla D_1(\bar{t}) \cdot \bar{dt}] + \gamma_2 [D_2^*(\bar{t}) + \nabla D_2(\bar{t}) \cdot \bar{dt}]$$

Take the derivatives of \mathcal{L} with respect to the components of \bar{dt} and set each equal to zero to obtain a stationary point:

$$\left. \begin{aligned} \frac{\partial \mathcal{L}}{\partial(dt_1)} &= 0 = 2 dt_1 + \gamma_1 \nabla D_1 \cdot \hat{t}_1 + \gamma_2 \nabla D_2 \cdot \hat{t}_1 \\ \frac{\partial \mathcal{L}}{\partial(dt_2)} &= 0 = 2 dt_2 + \gamma_1 \nabla D_1 \cdot \hat{t}_2 + \gamma_2 \nabla D_2 \cdot \hat{t}_2 \\ &\vdots \\ \frac{\partial \mathcal{L}}{\partial(dt_N)} &= 0 = 2 dt_N + \gamma_1 \nabla D_1 \cdot \hat{t}_N + \gamma_2 \nabla D_2 \cdot \hat{t}_N \end{aligned} \right\} \quad (A4)$$

where \hat{t}_i are unit vectors in the i^{th} direction. The arguments, \bar{t} , on D_1 and D_2 will presently be dropped.

Equation (A4) can be written as

$$\frac{\partial \mathcal{L}}{\partial(dt)} = 0 = 2 \bar{dt} + \gamma_1 \nabla D_1 + \gamma_2 \nabla D_2 \quad (A5)$$

Equations (A2) to (A4) form a set of $N + 2$ equations in $N + 2$ unknowns (dt_i and γ_1 and γ_2). To solve this, from equation (A5) obtain

$$\bar{dt} = -\frac{1}{2} (\gamma_1 \nabla D_1 + \gamma_2 \nabla D_2) \quad (A6)$$

Substitute equation (A6) into (A2) and (A3) and obtain

$$D_1^* + \nabla D_1 \cdot \left(-\frac{1}{2}\right)(\gamma_1 \nabla D_1 + \gamma_2 \nabla D_2) = 0 \quad (\text{A7})$$

and

$$D_2^* + \nabla D_2 \cdot \left(-\frac{1}{2}\right)(\gamma_1 \nabla D_1 + \gamma_2 \nabla D_2) = 0 \quad (\text{A8})$$

From equation (A8),

$$\gamma_1 = \frac{2D_2^* - \gamma_2 \nabla^2 D_2}{\nabla D_1 \cdot \nabla D_2} \quad (\text{A9})$$

Substitute equation (A9) into (A7) and solve for γ_2 to obtain

$$\gamma_2 = \frac{2D_1^*(\nabla D_1 \cdot \nabla D_2) - 2D_2^* \nabla^2 D_1}{(\nabla D_1 \cdot \nabla D_2)^2 - \nabla^2 D_1 \nabla^2 D_2} \quad (\text{A10})$$

Substitute equation (A10) into (A9) to obtain

$$\gamma_1 = \frac{2D_2^*(\nabla D_1 \cdot \nabla D_2) - 2D_1^* \nabla^2 D_2}{(\nabla D_1 \cdot \nabla D_2)^2 - \nabla^2 D_1 \nabla^2 D_2} \quad (\text{A11})$$

Finally, use equations (A6), (A10), and (A11) to obtain the value of $\overline{\text{dt}}$:

$$\overline{\text{dt}} = - \frac{[D_2^*(\nabla D_1 \cdot \nabla D_2) - D_1^* \nabla^2 D_2] \nabla D_1 + [D_1^*(\nabla D_1 \cdot \nabla D_2) - D_2^* \nabla^2 D_1] \nabla D_2}{(\nabla D_1 \cdot \nabla D_2)^2 - \nabla^2 D_1 \nabla^2 D_2} \quad (\text{A12})$$

APPENDIX B

FORTRAN IV LISTING

The FORTRAN-IV listing for the DOPEX-1D2C code is presented in this appendix. Included are the main routine and subroutines INPUT, INIT, CLEAR, DOZE, and WEIGHT. The coding is ANSI standard FORTRAN-IV and has been run successfully on both the IBM 7094-II and IBM 360/67 computers.

Description of Subroutines

The subroutines and their function in DOPEX-1D2C are as follows:

- (1) Main program: The main program calculates the steepest-descent vector and alters shield thicknesses; it calls subroutines INPUT, INIT, DOZE, WEIGHT, and CLEAR.
- (2) Subroutine INPUT: This subroutine reads all input data.
- (3) Subroutine INIT: If initial thicknesses do not result in the dose rates required by the constraints, INIT is called to calculate a set of thicknesses which does, so that the optimization can begin. This calculation is carried out without regard to weight minimization. INIT calls subroutines DOZE, WEIGHT, and CLEAR.
- (4) Subroutine WEIGHT: This subroutine calculates the total weight and the derivatives of weight with respect to thickness. It is presently coded for spherical geometry.
- (5) Subroutine DOZE: This subroutine calculates dose rates, the derivatives of dose rates with respect to thickness, and products of dose and weight derivatives.
- (6) Subroutine CLEAR: If a shield layer is eliminated in the course of a calculation, subroutine CLEAR is called to set that layer thickness to zero for the remainder of the calculation, to set the constraint flag so that it is maintained at zero thickness, and to zero out a dose component if it originated in the removed layer.

Program Listing

```
C      NEW ONE-DIMENSIONAL, DOUBLE CONSTRAINT OPTIMIZATION CODE
C      ...GPL 3 JAN 1972...  REVISED DEC 1972...
      COMMON MAX, EPS, EPSD, CON, CA, WT, GG, NREG,
1     DSTAR(2), T(25), RHO(25), NB(25), IMAX(2), D(25,2),
2     EMU(25,25,2), NIJ(25,2), DOSE(2), C(25,2), A(25,2),
3     AA(2), AXA, AG(2), U(25), G(25)
CCC
C      MAX          NUMBER OF ITERATIONS ALLOWED
C      EPS          PRECISION REQD ON WEIGHT CONVERGENCE
```



```

C      EPSD          PRECISION ON DOSE IN INITIALIZATION
C      CON           FRACTION FOR OPTIMIZATION UNIT VECTOR
C      CA            NOT USED PRESENTLY
C      WT            TOTAL SYSTEM WEIGHT
C      GG            G .DOT. G
C      DSTAR(2)      DOSE CONSTRAINTS
C      NREG          NUMBER OF REGIONS
C      T(25)         THICKNESS OF EACH REGION
C      RHO(25)       DENSITY OF EACH REGION
C      NB(25)        CONSTRAINT FLAG FOR EACH REGION
C      IMAX(2)       NUMBER OF DOSE COMPONENTS IN EACH DOSE CONSTRAINT
C      D(25,2)       VALUE OF THE RESPECTIVE DOSE COMPONENTS
C      EMU(25,25,2)  THE COEFFICIENTS FOR EACH REGION AND EACH CONSTRAINT
C      NIJ(25,2)     CROSS-REFERENCE-REGION NUMBER FOR EACH DOSE COMPONENT
C      DOSE(2)       PRESENT EVALUATION OF THE DOSE FOR EACH CONSTRAINT
C      C(25,2)       LEADING COEFFICIENT IN D(25,2) EQUATION
C      A(25,2)       D-D(1)/D-T(J)...I=1,2, J=1,NREG
C      AA(2)         DEL-SQUARED OF D(1)
C      AXA           DEL-D(1) .DOT. DEL-D(2)
C      AG(2)         DEL-D(1) .DOT. DEL-W
C      U(25)         COMPONENTS OF OPTIMIZATION UNIT VECTOR
C      G(25)         D-W/D-T(J)
C
5  CALL INPUT
   CALL WEIGHT
   CALL DOZE
   WRITE(6,7) WT
7  FORMAT(/20H0*** SYSTEM WEIGHT =,1PE12.4)
   IF((DSTAR(1) .GT. 0.0) .AND. (DSTAR(2) .GT. 0.0)) GO TO 10
   DSTAR(1)=DOSE(1)
   DSTAR(2)=DOSE(2)
   WRITE(6,8)(DSTAR(IC),IC=1,2)
8  FORMAT(44H0*** PROBLEM WILL OPTIMIZE TO CONSTRAINTS OF ,1P2E12.4)
   GO TO 12
C
10 CALL INIT
   WRITE(6,7) WT
C
12 WRITE(6,15)
15 FORMAT(40H1***** BEGIN OPTIMIZATION ***** )
C
C      MAIN PROGRAMME....
DO 100 K=1,MAX
  WTX=WT
  DET=AA(1)*AA(2)-AXA**2
  ALF1=(AG(2)*AXA-AG(1)*AA(2))/DET
  ALF2=(AG(1)*AXA-AG(2)*AA(1))/DET
  FORB2=GG+ALF1*(2.0*AG(1)+ALF1*AA(1)+2.0*ALF2*AXA)
  FORB2=FORB2+ALF2*(2.0*AG(2)+ALF2*AA(2))
C      THE ABOVE IS EQUATION(20) FROM NASA TM X-2435 4*BETA**2
  DENOM=SQRT(FORB2)
C
DO 20 I=1,NREG
  IF(NB(I) .EQ. 0) GO TO 20
  U(I)=(G(I)+ALF1*A(I,1)+ALF2*A(I,2))/DENOM
  T(I)=T(I)-CON*U(I)
  IF(T(I) .LE. 0.0) CALL CLEAR(I)
20 CONTINUE

```

```

C
C RESTORE DOSES TO CONSTRAINT
CALL DOZE
DENOM=AXA**2-AA(1)*AA(2)
D1=DOSE(1)-DSTAR(1)
D2=DOSE(2)-DSTAR(2)
GAMMA1=(D2*AXA-D1*AA(2))/DENOM
GAMMA2=(D1*AXA-D2*AA(1))/DENOM
DO 30 I=1,NREG
IF(NB(I).EQ. 0) GO TO 30
T(I)=T(I)-GAMMA1*A(I,1)-GAMMA2*A(I,2)
IF(T(I).LE. 0.0) CALL CLEAR(I)
30 CONTINUE

C
CALL WEIGHT
CALL DOZE

C
WRITE(6,40) K, WT, (DOSE(IC), IC=1,2)
40 FORMAT(8H0*** K =,I3,5X,12H*** WEIGHT =,1PE12.4,
1 5X,17H*** TOTAL DOSES =,2E12.4)
WRITE(6,42)(T(I), I=1,NREG)
42 FORMAT(9H T = ,1P10E12.4/(9X,10E12.4))

C
IF((K/5)*5.NE. K) GO TO 55
ISLIP=1
45 DO 48 IC=1,2
IMMX=IMAX(IC)
48 WRITE(6,49) IC,(D(I,IC),I=1,IMMX)
49 FORMAT(42H0 DOSE COMPONENTS FOR CONSTRAINT NUMBER,I2/
1 (9X,1P10E12.4))
GO TO(55,5),ISLIP

C
55 IF(ABS((WTX-WT)/WT).GT. EPS) GO TO 100
ISLIP=2
GO TO 45

C
100 CONTINUE
STOP
END

```

SUBROUTINE INPUT

```

C NEW ONE-DIMENSIONAL, DOUBLE CONSTRAINT OPTIMIZATION CODE
C ...GPL 3 JAN 1972... REVISED DEC 1972..
COMMON MAX, EPS, EPSD, CON, CA, WT, GG, NREG,
1 DSTAR(2), T(25), RHO(25), NB(25), IMAX(2), D(25,2),
2 EMU(25,25,2), NIJ(25,2), DOSE(2), C(25,2), A(25,2),
3 AA(2), AXA, AG(2), U(25), G(25)

CCC
C MAX NUMBER OF ITERATIONS ALLOWED
C EPS PRECISION REQD ON WEIGHT CONVERGENCE
C EPSD PRECISION ON DOSE IN INITIALIZATION
C CON FRACTION FOR OPTIMIZATION UNIT VECTOR
C IC CONSTRAINT COUNTER
C DSTAR(2) DOSE CONSTRAINTS
C NREG NUMBER OF REGIONS
C T(25) THICKNESS OF EACH REGION
C RHO(25) DENSITY OF EACH REGION

```

```

C      NB(25)          CONSTRAINT FLAG FOR EACH REGION
C      IMAX(2)         NUMBER OF DOSE COMPONENTS IN EACH DOSE CONSTRAINT
C      D(25,2)         VALUE OF THE RESPECTIVE DOSE COMPONENTS
C      EMU(25,25,2)    THE COEFFICIENTS FOR EACH REGION AND EACH CONSTRAINT
C      NIJ(25,2)       CROSS-REFERENCE-REGION NUMBER FOR EACH DOSE COMPONENT
C      DOSE(2)         PRESENT EVALUATION OF THE DOSE FOR EACH CONSTRAINT
C      C(25,2)         LEADING COEFFICIENT IN D(25,2) EQUATION
C
C      DIMENSION ALPHA(20)
C
C      READ(5,5) ALPHA
C      5 FORMAT(20A4)
C      WRITE(6,10) ALPHA
C      10 FORMAT(1H1,5X,20A4)
C
C      READ(5,15) NREG, MAX, EPS, EPSD, CON
C      15 FORMAT(2I5, 4E10.4)
C      WRITE(6,20) NREG, MAX, EPS, EPSD, CON
C      20 FORMAT(8H0 NREG =,13,5X,5HMAX =,14,5X,5HEPS =,F8.5,5X,
C      1 6HEPSD =,F8.5,5X,5HCON =,F8.3,5X)
C
C      READ(5,30) (T(I), I=1,NREG)
C      30 FORMAT(7E10.0)
C      READ(5,30) (RHO(I), I=1,NREG)
C      READ(5,35) (NB(I), I=1,NREG)
C      35 FORMAT(25I1)
C      WRITE(6,36)(I, T(I), RHO(I), NB(I), I=1,NREG)
C      36 FORMAT(/,34H0REGION      T(I)      RHO(I)  NB(I)/(17,2F10.3,17))
C
C      DO 70 IC=1,2
C      READ(5,37) IMAX(IC), DSTAR(IC)
C      37 FORMAT(15,E10.4)
C      WRITE(6,38) IC, IMAX(IC), DSTAR(IC)
C      38 FORMAT(9H1****IC =,12,5X,6HIMAX =,13,5X,12HCONSTRAINT =,1PE12.4)
C      IMMX=IMAX(IC)
C
C      WRITE(6,39)
C      39 FORMAT(27H0REGION-J      MU(1,J))
C      DO 40 J=1,NREG
C      READ(5,30) (EMU(I,J,IC), I=1,IMMX)
C      40 WRITE(6,45) J, (EMU(I,J,IC), I=1,IMMX)
C      45 FORMAT(16, 3X, 1P9E12.4/(9X,1P9E12.4))
C
C      READ(5,30)(D(I,IC), I=1,IMMX)
C      READ(5,50)(NIJ(I,IC), I=1,IMMX)
C      50 FORMAT(14I5)
C
C      DOSE(IC)=0.0
C      DO 60 I=1,IMMX
C      DOSE(IC)=DOSE(IC)+D(I,IC)
C      BB=0.0
C      DO 55 J=1,NREG
C      55 BB=BB+EMU(I,J,IC)*T(J)
C      60 C(I,IC)=D(I,IC)*EXP(BB)
C      WRITE(6,65)(I,C(I,IC),D(I,IC),NIJ(I,IC), I=1,IMMX)

```

```

65 FORMAT(1H0/34H0   I           C(I)           D(I)  NIJ/(15,1P2E12.4,15))
   WRITE(6,68) DOSE(IC)
68 FORMAT(16H0   TOTAL DOSE =,1PE12.4)
C
70 CONTINUE
100 RETURN
   END

SUBROUTINE INIT
C \ NEW ONE-DIMENSIONAL, DOUBLE CONSTRAINT OPTIMIZATION CODE
C ...GPL 3 JAN 1972...
C ...REVISED GPL 18 DEC 1972...
COMMON MAX, EPS, EPSD, CON, CA, WT, GG, NREG,
1 DSTAR(2), T(25), RHO(25), NB(25), IMAX(2), D(25,2),
2 EMU(25,25,2), NIJ(25,2), DOSE(2), C(25,2), A(25,2),
3 AA(2), AXA, AG(2), U(25), G(25)
CCC
   DIMENSION DEL(2)
C
1 FACTOR=0.5
  FLOW=FACTOR
  FHI=1.0/FACTOR
C
C
DO 67 K=1,MAX
C
DO 55 IC=1,2
  IF(DOSE(IC)-FHI*DSTAR(IC)) 52,52,51
51 DEL(IC)=DOSE(IC)*(1.0-FLOW)
  DOSE(IC) IS TOO HIGH...ADJUST T CORRESPONDING TO
  A DOSE CONSTRAINT OF FLOW*DOSE(IC)
  GO TO 55
C
52 IF(DOSE(IC)-FLOW*DSTAR(IC)) 53,53,54
53 DEL(IC)=DOSE(IC)*(1.0-FHI)
  DOSE IS TOO LOW...ADJUST T CORRESPONDING TO
  A DOSE CONSTRAINT OF FHI*DOSE(IC)
  GO TO 55
C
54 DEL(IC)=DOSE(IC)-DSTAR(IC)
  DOSE(IC)/DSTAR(IC) IS .GT. FACTOR .AND. .LT. 1/FACTOR
  ASSUME FIRST ORDER CORRECTION MODEL TO WORK
55 CONTINUE
C
  DENOM=AXA**2-AA(1)*AA(2)
  GAMMA1=(DEL(2)*AXA-DEL(1)*AA(2))/DENOM
  GAMMA2=(DEL(1)*AXA-DEL(2)*AA(1))/DENOM
C
  U(I)=GAMMA1*A(I,1)+GAMMA2*A(I,2)
C
DO 66 I=1,NREG
  IF( NB(I) .EQ. 0) GO TO 66
  U(I)=GAMMA1*A(I,1)+GAMMA2*A(I,2)
  T(I)=T(I)-U(I)
  IF(T(I) .LE. 0.0) CALL CLEAR(I)
66 CONTINUE
C
C TEST CONVERGENCE
  CALL DOZE

```

```

      IF(ABS(DOSE(1)/DSTAR(1) -1.0) .GT. EPSD) GO TO 67
      IF(ABS(DOSE(2)/DSTAR(2) -1.0) .LT. EPSD) GO TO 69
67 CONTINUE
      K=777

C
C  UPDATE ALL DERIVATIVES
69 CALL WEIGHT
   CALL DOZE
   WRITE(6,70) K, WT, (DOSE(IC), IC=1,2)
   WRITE(6,75)(T(I), I=1,NREG)
   DO 80 IC=1,2
     IMX=IMAX(IC)
80  WRITE(6,85) IC,(D(I,IC), I=1,IMX)
     RETURN
70  FORMAT(28H1***INITIALIZATION REQUIRED ,13,12H ITERATIONS./
1    12H0  WEIGHT =,1PE12.4/
2    12H0  DOSES ARE ,1PE12.4)
75  FORMAT(32H0**** INITIAL THICKNESSES ARE.../(1P10E12.4))
85  FORMAT(41H0**** INITIAL DOSES FOR CONSTRAINT NUMBER,12/
1    (1P10E12.4))
     END

      SUBROUTINE DOZE
C      NEW ONE-DIMENSIONAL, DOUBLE CONSTRAINT OPTIMIZATION CODE
C      ...GPL 3 JAN 1972...
      COMMON MAX, EPS, EPSD, CON, CA, WT, GG, NREG,
1  DSTAR(2), T(25), RHO(25), NB(25), IMAX(2), D(25,2),
2  EMU(25,25,2), NIJ(25,2), DOSE(2), C(25,2), A(25,2),
3  AA(2), AXA, AG(2), U(25), G(25)

CCC
      DO 30 IC=1,2
      DOSE(IC)=0.0
      IMX=IMAX(IC)
      DO 10 I=1,IMX
      BB=0.0
      DO 5 J=1,NREG
5  BB=BB+EMU(I,J,IC)*T(J)
      D(I,IC)=C(I,IC)*EXP(-BB)
10  DOSE(IC)=DOSE(IC)+D(I,IC)

C
      AA(IC)=0.0
      AG(IC)=0.0
      DO 20 K=1,NREG
      A(K,IC)=0.0
      IF(NB(K) .EQ. 0) GO TO 20
      DO 15 I=1,IMX
15  A(K,IC)=A(K,IC)-EMU(I,K,IC)*D(I,IC)
      AA(IC)=AA(IC)+A(K,IC)**2

C ***
C *** MAKE SURE THAT YOU HAVE CURRENT VALUES OF AG AND G
C *** WHEN EVALUATING THIS NEXT TERM.....
      AG(IC)=AG(IC)+A(K,IC)*G(K)
20 CONTINUE
C

```

```

30 CONTINUE
C
  AXA=0.0
  DO 40 K=1,NREG
40 AXA=AXA+A(K,1)*A(K,2)
C
  RETURN
  END

SUBROUTINE WEIGHT
C NEW ONE-DIMENSIONAL, DOUBLE CONSTRAINT OPTIMIZATION CODE
C ...GPL 3 JAN 1972...
  COMMON MAX, EPS, EPSD, CON, CA, WT, GG, NREG,
  1 DSTAR(2), T(25), RHO(25), NB(25), IMAX(2), D(25,2),
  2 EMU(25,25,2), NIJ(25,2), DOSE(2), C(25,2), A(25,2),
  3 AA(2), AXA, AG(2), U(25), G(25)
CCC DATA FORPI/12.56636/
C
C SPHERICAL GEOMETRY
C DIMENSION R(26)
C
  R(1)=T(1)
  RRR=R(1)**3
  WT=RHO(1)*RRR
  DO 5 I=2,NREG
  R3=RRR
  R(I)=R(I-1)+T(I)
  RRR=R(I)**3
  WT=WT+RHO(I)*(RRR-R3)
5 CONTINUE
  WT=WT*FORPI/3.0

C
C CALCULATE PARTIAL WEIGHT DERIVATIVES
C
  GG=0.0
  DO 7 I=1,NREG
  RR=0.0
  G(I)=0.0
  IF(NB(I) .EQ. 0) GO TO 7
  DO 6 J=1,NR
  R2=RR
  RR=R(J)**2
6 G(I)=G(I)+RHO(J)*(RR-R2)
  G(I)=G(I)*FORPI
  GG=GG+G(I)**2
7 CONTINUE
  RETURN
  END

```

```

SUBROUTINE CLEAR(J)
C NEW ONE-DIMENSIONAL, DOUBLE CONSTRAINT OPTIMIZATION CODE
C ...GPL 3 JAN 1972...
COMMON MAX, EPS, EPSD, CON, CA, WT, GG, NREG,
1 DSTAR(2), T(25), RHO(25), NB(25), IMAX(2), D(25,2),
2 EMU(25,25,2), NIJ(25,2), DOSE(2), C(25,2), A(25,2),
3 AA(2), AXA, AG(2), U(25), G(25)
CCC
C
C THE JTH REGION HAS JUST BEEN WIPED OUT
C
T(J)=0.0
NB(J)=0
DO 5 IC=1,2
IM=IMAX(IC)
DO 5 I=1,IM
IF(J .EQ. NIJ(I,IC)) C(I,IC)=0.0
5 CONTINUE
RETURN
END

```

REFERENCES

1. Lahti, Gerald P.; and Herrmann, Paul F.: Comparison of Tungsten and Depleted Uranium in Minimum-Weight, Layered Shields for a Space Power Reactor. NASA TM X-1874, 1969.
2. Engle, W. W., Jr.; and Mynatt, F. R.: A Shield Optimization Technique with Direct Utilization of Transport Calculations. Trans. ANS, vol. 12, no. 2, Dec. 1969, p. 953.
3. Bernick, R. L.: Application of the Method of Steepest Descent to Laminated Shield Weight Optimization. Rep. NAA-SR-Memo-8181, North American Aviation, Inc., Apr. 17, 1963.
4. Bernick, R. L.: The OPEX Shield Optimization Code. Rep. NAA-SR-Memo-11516, North American Aviation, Inc., July 13, 1965.
5. Lahti, Gerald P.: OPEX-II, A Radiation Shield Optimization Code. NASA TM X-1769, 1969.
6. Lahti, Gerald P.: Application of the Method of Steepest Descent to Laminated Shield Weight Optimization with Several Constraints - Theory. NASA TM X-2435, 1971.
7. Lahti, Gerald P.: The DOPEX Code - An Application of the Method of Steepest Descent to Laminated-Shield Weight Optimization with Several Constraints. NASA TM X-2554, 1972.

TABLE I. - REGION DESCRIPTION

Region, j	Description	Density, g/cm ³	Initial thickness, cm (guess)	Final prediction of shield thicknesses by OPEX-II, cm	Final prediction of shield thicknesses by DOPEX-1D2C, cm
1	Reactor core	9.957	26.0 radius	-----	-----
2	Plenum	8.647	2.50	-----	-----
3	Pressure vessel	16.763	.60	-----	-----
4	Molybdenum reflector	9.234	11.00	-----	-----
5	Lithium hydride	.75	(17.90)	20.52	22.49
6	Tungsten	19.3	(7.00)	9.71	9.62
7	Lithium hydride	.75	(14.00)	12.32	12.75
8	Tungsten	19.3	(5.00)	2.82	2.99
9	Lithium hydride	.75	(10.00)	10.32	9.51
10	Tungsten	19.3	(3.50)	2.33	1.81
11	Lithium hydride	.75	(59.50)	39.29	38.24

TABLE II. - DOSE COMPONENTS

Item, i	Dose component, D _i	Initial shield thicknesses	Final shield thicknesses	
			OPEX-II	DOPEX-1D2C
		Value of dose component D _i , mrem/hr		
1	Neutron	0.02430	0.2527	0.2527
2	Core gamma	.00303	.0079	.0110
3	Plenum, pressure vessel capture gamma	.00196	.0049	.0069
4	Plenum, pressure vessel inelastic gamma	.00220	.0055	.0077
5	Reflector capture gamma	.204	.4782	.6710
6	Reflector inelastic gamma	.00504	.0129	.0178
7	Region 6 tungsten capture gamma	.0921	.4007	.2703
8	Region 6 tungsten inelastic gamma	.00974	.0949	.0842
9	Region 8 tungsten capture gamma	.0988	.2696	.2590
10	Region 8 tungsten inelastic gamma	.0278	.0759	.0788
11	Region 10 tungsten capture gamma	.201	.2650	.2332
12	Region 10 tungsten inelastic gamma	.0947	.1320	.1073
Total		0.7647	2.000	2.000

TABLE III. - SAMPLE DOPEX-1D2C OUTPUT

***TEST THE NEW DOUBLE CONSTRAINT DOPEX * * *

NREG = 11 MAX = 50 EPS = 0.00100 EPSD = 0.00700 CON = 1.000

REGION	T(I)	RHO(I)	NR(I)
1	26.000	9.957	0
2	2.500	8.647	0
3	0.600	16.763	0
4	11.000	9.234	0
5	17.900	0.750	1
6	7.000	19.300	1
7	14.000	0.750	1
8	5.000	19.300	1
9	10.000	0.750	1
10	3.500	19.300	1
11	59.500	0.750	1

****IC = 1 IMAX = 1 CONSTRAINT = 2.5270E-01

REGION-J	MU(I,J)
1	0.0000
2	0.0000
3	0.0000
4	0.0000
5	1.3470E-01
6	2.4800E-01
7	1.3690E-01
8	2.3160E-01
9	1.2180E-01
10	2.3060E-01
11	1.1870E-01

I	C(I)	D(I)	NIJ
1	2.9425E 05	2.4300E-02	1

TOTAL DOSE = 2.4300E-02

****IC = 2 I MAX = 11 CONSTRAINT = 1.7473E 00

REGION-J	MU(I,J)										
1	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000
2	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000
3	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000
4	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000
5	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000	0.0000
6	1.8900E-02	1.4700E-02	1.4700E-02	2.1200E-02	2.4000E-02	3.6290E-01	2.2970E-01	2.0113E-01	1.8850E-01	1.8850E-01	1.8850E-01
7	1.8370E-01	1.7590E-01	7.7200E-01	8.0500E-01	7.9503E-01	2.7190E-01	1.8120E-01	2.3860E-01	2.5360E-01	2.5360E-01	2.5360E-01
8	7.7400E-01	2.5770E-01	2.0400E-02	1.6700E-02	2.0600E-02	2.1000E-02	2.5000E-02	2.2570E-01	1.9830E-01	1.9830E-01	1.9830E-01
9	2.5430E-01	2.0400E-02	1.7950E-01	7.9000E-01	7.8100E-01	7.9800E-01	8.2700E-01	1.7793E-01	1.2550E-01	1.2550E-01	1.2550E-01
10	2.2800E-02	2.0400E-02	2.2000E-02	2.0100E-02	2.3900E-02	1.8700E-02	2.4500E-02	2.3000E-02	2.5000E-02	2.5000E-02	2.5000E-02
11	1.9100E-01	7.6600E-01	7.6300E-01	8.0500E-01	7.7600E-01	7.8200E-01	8.1200E-01	8.1000E-01	8.5100E-01	8.5100E-01	8.5100E-01
	2.0520E-01	2.3680E-01	2.2000E-02	2.2600E-02	2.5200E-02	2.3200E-02	2.8600E-02	2.3200E-02	2.9100E-02	2.9100E-02	2.9100E-02
	2.2310E-01	1.8080E-01	7.6300E-01	1.1070E-01	7.6300E-01	7.8200E-01	8.1200E-01	8.1000E-01	8.5100E-01	8.5100E-01	8.5100E-01
	7.6800E-01	7.6300E-01	1.1070E-01	2.2600E-02	2.5200E-02	2.3200E-02	2.8600E-02	2.3200E-02	2.9100E-02	2.9100E-02	2.9100E-02
	2.4070E-01	1.1070E-01	2.2600E-02	2.2600E-02	2.5200E-02	2.3200E-02	2.8600E-02	2.3200E-02	2.9100E-02	2.9100E-02	2.9100E-02
	2.3100E-02	2.9700E-02	2.9700E-02	2.9700E-02	2.9700E-02	2.9700E-02	2.9700E-02	2.9700E-02	2.9700E-02	2.9700E-02	2.9700E-02

I	C(I)	D(I)	NIJ
1	4.9449E 03	3.0300E-03	1
2	2.3997E 03	1.9600E-03	2
3	2.6936E 03	2.2000E-03	3
4	3.5981E 05	2.0400E-01	4
5	1.1522E 04	5.0400E-03	4
6	2.1972E 05	9.2100E-02	5
7	2.2521E 04	9.7400E-03	6
8	9.5078E 04	9.8800E-02	8
9	2.0543E 04	2.7800E-02	8
10	1.1039E 05	2.0100E-01	10
11	2.8426E 04	9.4700E-02	10
TOTAL DOSE =			7.4037E-01

*** SYSTEM WEIGHT = 3.5939E 07

TABLE III. - Concluded. SAMPLE DOPEX-1D2C OUTPUT

***INITIALIZATION REQUIRED 5 ITERATIONS.

WEIGHT = 3.2451E 07

DOSES ARE 2.5270E-01 1.7485E 00

*** INITIAL THICKNESSES ARE...

2.6000E 01 2.5000E 00 6.0000E-01 1.1000E 01 1.6078E 01 5.0377E 00 1.0587E 01 4.6848E 00 8.7432E 00 6.4919E 00
 4.6031E 01

*** INITIAL DOSES FOR CONSTRAINT NUMBER 1

2.5270E-01

*** INITIAL DOSES FOR CONSTRAINT NUMBER 2

2.9056E-03 1.8187E-03 2.0414E-03 1.7198E-01 4.9993E-03 5.7133E-02 4.0084E-03 6.5402E-02 1.5917E-02 8.3939E-01
 5.8291E-01

*** SYSTEM WEIGHT = 3.2451E 07

***** BEGIN OPTIMIZATION *****

*** K = 1 *** WEIGHT = 3.1811E 07 *** TOTAL DOSES = 2.5270E-01 1.7492E 00

T = 2.6000E 01 2.5000E 00 6.0000E-01 1.1000E 01 1.6482E 01 5.2073E 00 1.0943E 01 4.5011E 00 9.1257E 00 5.3141E 00
 4.5832E 01

*** K = 2 *** WEIGHT = 3.1365E 07 *** TOTAL DOSES = 2.5270E-01 1.7488E 00

T = 2.6000E 01 2.5000E 00 6.0000E-01 1.1000E 01 1.6884E 01 5.4781E 00 1.1244E 01 4.3677E 00 9.1785E 00 5.3970E 00
 4.5367E 01

*** K = 3 *** WEIGHT = 3.1047E 07 *** TOTAL DOSES = 2.5270E-01 1.7479E 00

T = 2.6000E 01 2.5000E 00 6.0000E-01 1.1000E 01 1.7274E 01 5.8236E 00 1.1477E 01 4.2742E 00 9.7453E 00 4.9170E 00
 4.4775E 01

*** K = 4 *** WEIGHT = 3.0804E 07 *** TOTAL DOSES = 2.5270E-01 1.7475E 00

T = 2.6000E 01 2.5000E 00 6.0000E-01 1.1000E 01 1.7668E 01 6.2169E 00 1.1659E 01 4.1977E 00 9.9270E 00 4.4655E 00
 4.4135E 01

*** K = 5 *** WEIGHT = 3.0616E 07 *** TOTAL DOSES = 2.5270E-01 1.7474E 00

T = 2.6000E 01 2.5000E 00 6.0000E-01 1.1000E 01 1.8084E 01 6.6302E 00 1.1812E 01 4.1193E 00 1.0038E 01 4.0475E 00
 4.3473E 01

DOSE COMPONENTS FOR CONSTRAINT NUMBER 1

2.5270E-01

DOSE COMPONENTS FOR CONSTRAINT NUMBER 2

8.0735E-03 5.0440E-03 5.6616E-03 5.0090E-01 1.3702E-02 1.9020E-01 2.2121E-02 1.3532E-01 5.0803E-02 4.9879E-01
2.7679E-01

*** K = 6 *** WEIGHT = 3.0469E 07 *** TOTAL DOSES = 2.5270E-01 1.7474E 00
T = 2.6000E 01 2.5000E 00 6.0000E-01 1.1000E 01 1.8532E 01 7.0439E 00 1.1950E 01 4.0285E 00 1.0093E 01 3.6676E 00
4.2801E 01

*** K = 7 *** WEIGHT = 3.0355E 07 *** TOTAL DOSES = 2.5270E-01 1.7473E 00
T = 2.6000E 01 2.5000E 00 6.0000E-01 1.1000E 01 1.9018E 01 7.4469E 00 1.2081E 01 3.9187E 00 1.0101E 01 3.3254E 00
4.2126E 01

*** K = 8 *** WEIGHT = 3.0267E 07 *** TOTAL DOSES = 2.5270E-01 1.7473E 00
T = 2.6000E 01 2.5000E 00 6.0000E-01 1.1000E 01 1.9540E 01 7.8353E 00 1.2206E 01 3.7904E 00 1.0070E 01 3.0201E 00
4.1455E 01

*** K = 9 *** WEIGHT = 3.0197E 07 *** TOTAL DOSES = 2.5270E-01 1.7473E 00
T = 2.6000E 01 2.5000E 00 6.0000E-01 1.1000E 01 2.0090E 01 8.2099E 00 1.2328E 01 3.5559E 00 1.0006E 01 2.7425E 00
4.0792E 01

*** K = 10 *** WEIGHT = 3.0142E 07 *** TOTAL DOSES = 2.5270E-01 1.7473E 00
T = 2.6000E 01 2.5000E 00 6.0000E-01 1.1000E 01 2.0665E 01 8.5733E 00 1.2443E 01 3.4921E 00 9.9152E 00 2.4871E 00
4.0139E 01

DOSE COMPONENTS FOR CONSTRAINT NUMBER 1

2.5270E-01

DOSE COMPONENTS FOR CONSTRAINT NUMBER 2

9.8503E-03 6.1521E-03 6.9054E-03 6.0595E-01 1.6215E-02 2.6255E-01 5.5713E-02 2.5745E-01 7.5952E-02 3.0246E-01
1.4810E-01

*** K = 11 *** WEIGHT = 3.0099E 07 *** TOTAL DOSES = 2.5270E-01 1.7473E 00
T = 2.6000E 01 2.5000E 00 6.0000E-01 1.1000E 01 2.1259E 01 8.9280E 00 1.2552E 01 3.1295E 00 9.8013E 00 2.2487E 00
3.9496E 01

*** K = 12 *** WEIGHT = 3.0065E 07 *** TOTAL DOSES = 2.5270E-01 1.7473E 00
T = 2.6000E 01 2.5000E 00 6.0000E-01 1.1000E 01 2.1868E 01 9.2760E 00 1.2653E 01 3.1620E 00 9.5579E 00 2.0234E 00
3.8863E 01

*** K = 13 *** WEIGHT = 3.0040E 07 *** TOTAL DOSES = 2.5270E-01 1.7473E 00
T = 2.6000E 01 2.5000E 00 6.0000E-01 1.1000E 01 2.2490E 01 9.6187E 00 1.2746E 01 2.9911E 00 9.5177E 00 1.8085E 00
3.8239E 01

DOSE COMPONENTS FOR CONSTRAINT NUMBER 1

2.5270E-01

DOSE COMPONENTS FOR CONSTRAINT NUMBER 2

1.1021E-02 5.8893E-03 7.7329E-03 6.7103E-01 1.7816E-02 2.7035E-01 8.4234E-02 2.5902E-01 7.9795E-02 2.3318E-01
1.0725E-01

TABLE IV. - COEFFICIENTS USED IN SAMPLE PROBLEM

Region, j	Description	Dose component, D _i (i = 1 to 12)											
		1	2	3	4	5	6	7	8	9	10	11	12
		Neutron	Core gamma	Plenum n, γ	Plenum n, n'γ	Reflector n, γ	Reflector n, n'γ	W(6) n, γ	W(6) n, n'γ	W(8) n, γ	W(8) n, n'γ	W(10) n, γ	W(10) n, n'γ
Attenuation coefficient, μ _{ij} , cm ⁻¹													
1	Core	0	0	0	0	0	0	0	0	0	0	0	0
2	Plenum	0	0	0	0	0	0	0	0	0	0	0	0
3	Pressure vessel	0	0	0	0	0	0	0	0	0	0	0	0
4	Reflector	0	0	0	0	0	0	0	0	0	0	0	0
5	Lithium hydride	.1347	.0189	.0147	.0147	.0212	.024	.3629	.2297	.2018	.1885	.1837	.1759
6	Tungsten	.2480	.774	.772	.772	.805	.795	.2719	.1812	.2386	.2536	.2543	.2577
7	Lithium hydride	.1369	.0228	.0204	.0204	.0167	.0206	.0210	.0250	.2257	.1983	.1910	.1795
8	Tungsten	.2316	.770	.766	.766	.790	.781	.798	.827	.1779	.1255	.2052	.2368
9	Lithium hydride	.1218	.024	.022	.022	.0201	.0239	.0187	.0245	.0230	.0250	.2231	.1808
10	Tungsten	.2306	.768	.763	.763	.805	.776	.782	.812	.810	.851	.2407	.1107
11	Lithium hydride	.1187	.0244	.0226	.0226	.0196	.0252	.0232	.0286	.0232	.0291	.0231	.0297




Figure 1. - Geometry for sample problem.



POSTMASTER: If Undeliverable (Section 158, Postal Manual) Do Not Return

"The aeronautical and space activities of the United States shall be conducted so as to contribute . . . to the expansion of human knowledge of phenomena in the atmosphere and space. The Administration shall provide for the widest practicable and appropriate dissemination of information concerning its activities and the results thereof."

—NATIONAL AERONAUTICS AND SPACE ACT OF 1958

NASA SCIENTIFIC AND TECHNICAL PUBLICATIONS

TECHNICAL REPORTS: Scientific and technical information considered important, complete, and a lasting contribution to existing knowledge.

TECHNICAL NOTES: Information less broad in scope but nevertheless of importance as a contribution to existing knowledge.

TECHNICAL MEMORANDUMS: Information receiving limited distribution because of preliminary data, security classification, or other reasons. Also includes conference proceedings with either limited or unlimited distribution.

CONTRACTOR REPORTS: Scientific and technical information generated under a NASA contract or grant and considered a significant contribution to existing knowledge.

TECHNICAL TRANSLATIONS: Information published in a foreign language considered to merit NASA distribution in English.

SPECIAL PUBLICATIONS: Information derived from or of value to NASA activities. Publications include final reports of major projects, monographs, data compilations, handbooks, sourcebooks, and special bibliographies.

TECHNOLOGY UTILIZATION PUBLICATIONS: Information on technology used by NASA that may be of particular interest in commercial and other non-aerospace applications. Publications include Tech Briefs, Technology Utilization Reports and Technology Surveys.

Details on the availability of these publications may be obtained from:

SCIENTIFIC AND TECHNICAL INFORMATION OFFICE
NATIONAL AERONAUTICS AND SPACE ADMINISTRATION
Washington, D.C. 20546